TubeSat Power Estimates

Estimate the solar power available to a TubeSat in a 310 km sun synchronous polar orbit.

version 2011-04-18
by Wesley Faler

Dimensions and Constants

In[1]:=
```
photoWidth = 0.02109; (* width of photocell panel, meters *)
photoRadius = 0.04343; (* distance from center to face of photocell panel, meters *)
photoAngle = 27.295/360 2 π; (* angle of circle for one photocell panel, radians *)
alAngle = 17.632/360 2 π; (* angle of circle for one aluminum strip, radians *)
photoLength = 0.127; (* length of the photocell panel, meters *)
wattPerPanel = 2.277 * 0.27 * 0.1336 * 6;
(* 2.277 cm^2/cell * 0.27 efficiency * 0.1366 W/cm^2 * 6 cells/panel *)
halfOrbitSeconds = 45 * 60; (* seconds for half of an orbit at 310 km *)
```

Panels and Normal Vectors

The center of the TubeSat is at {0, 0, 0}.
The sun in the + X direction.
panelPoints holds 3D locations of the solar panel centers, all in the XY plane.
normalPoints holds 3D locations of a unit vector normal to each solar panel.
{normalPoints - panelPoints} are the normal vector of each solar panel.

In[44]:=
```
Angle[plateIndex_] = plateIndex * (photoAngle + alAngle);
MakePoint[plateIndex_, x_, y_, z_] = {
   x Cos[Angle[plateIndex]] - y Sin[Angle[plateIndex]],
   y Cos[Angle[plateIndex]] + x Sin[Angle[plateIndex]],
   z};
MakeCell[plateIndex_, y_, z_] = {
   MakePoint[plateIndex, photoRadius, y, z],
   MakePoint[plateIndex, photoRadius + 1, y, z]};
panelPoints = Table[MakePoint[i, photoRadius, 0, 0], {i, 0, 7}];
normalPoints = Table[MakePoint[i, photoRadius + 1, 0, 0], {i, 0, 7}];
```

Rotate Panels and Calculate Solar Cosines

Rotate panelPoints and normalPoints about the TubeSat center.
Create the normal vector for each panel and dot product it with the sun’s location at inifnite +X.
Sum the positive dot products - these are the panels that get any sunlight. These must still be multiplied by the power per solar panel and then integrated over time as the satellite orbits.

In[12]:=
```
R[p_, xa_, ya_, za_] = p.RotationMatrix[xa Degree, {1, 0, 0}].
   RotationMatrix[ya Degree, {0, 1, 0}].RotationMatrix[za Degree, {0, 0, 1}];
(* rotate a point around each axis *)
PositivePower[p_] = If[p > 0, p, 0];
(* clip power factors to postive values only. panels facing away
 from the sun have zero contribution, not negative contribution. *)
PowerFactors[xa_, ya_, za_] =
  Map[PositivePower, (R[normalPoints, xa, ya, za] - R[panelPoints, xa, ya, za]).{1, 0, 0}];
PowerFactor[xa_, ya_, za_] = Total[PowerFactors[xa, ya, za]];
```

Power Factor for Rotation about X Axis

In[16]:=
```
Plot[PowerFactor[xa, 0, 0], {xa, 0, 360}]
PowerFactor[0, 0, 0]
```

Out[16]=



Out[17]= 2.41133

Power Factor for Rotation about Y Axis

In[18]:=
```
Plot[PowerFactor[0, ya, 0], {ya, 0, 360}]
```

Out[18]=



Power Factor for Rotation about Z Axis

In[19]:=
```
Plot[PowerFactor[0, 0, za], {za, 0, 360}]
```

Out[19]=



Flat Orientation

“Flat” Orientation: nose toward sun at one pole, tail to sun at the other, broadside to sun at equator. Use this if the thrust line is the long axis of the satellite and the thrust line is always kept in the direction of travel.
2042 Joules are taken in during the half orbit facing the sun. Over the half orbit period, this equates to an average of 0.75 Watts.

In[20]:=
$$\text{OrbitAngleFlat[t\_]} = \left(\frac{t}{\text{halfOrbitSeconds}}\right) * 180 - 90;$$

```
Plot[PowerFactor[0, OrbitAngleFlat[t], 0] * wattPerPanel, {t, 0, halfOrbitSeconds}]
```

$$\text{energyHalfOrbitFlat} = \text{wattPerPanel} \int_0^{\text{halfOrbitSeconds}} \text{PowerFactor[0, OrbitAngleFlat[t], 0]}\, dt$$

```
(* joules *)
```

$$\text{averagePowerHalfOrbitFlat} = \frac{\text{energyHalfOrbitFlat}}{\text{halfOrbitSeconds}} \text{ (* watts *)}$$

Out[21]=



Out[22]= 2042.61

Out[23]= 0.756521

Radial Orientation

“Radial” Orientation: broadside to sun at each pole, endcap to sun at equator. Use this if a narrow end should always point toward Earth.
2047 Joules are taken in during the half orbit facing the sun. Over the half orbit period, this equates to an average of 0.75 Watts.

```
In[24]:= OrbitAngleRadial[t_] = (t / halfOrbitSeconds) * 180;
Plot[PowerFactor[0, OrbitAngleRadial[t], 0] * wattPerPanel, {t, 0, halfOrbitSeconds}]
energyHalfOrbitRadial = wattPerPanel Integrate[PowerFactor[0, OrbitAngleRadial[t], 0], {t, 0, halfOrbitSeconds}]
(* joules *)
averagePowerHalfOrbitRadial = energyHalfOrbitRadial / halfOrbitSeconds (* watts *)
```

Out[25]=



Out[26]= 2047.82

Out[27]= 0.75845

Sun Seeker Orientation

“SunSeeker” Orientation: broadside always to the sun, likely rotating along long axis for cooling.
3208 Joules are taken in during the half orbit facing the sun. Over the half orbit period, this equates to an average of 1.18 Watts.

In[28]:=
```
OrbitAngleSunSeeker[t_] = 0;
Plot[PowerFactor[0, OrbitAngleSunSeeker[t], 0] * wattPerPanel, {t, 0, halfOrbitSeconds}]
energyHalfOrbitSunSeeker =
```

$$\text{wattPerPanel} \int_0^{\text{halfOrbitSeconds}} \text{PowerFactor}[0, \text{OrbitAngleSunSeeker}[t], 0]\, dt \quad \text{(* joules *)}$$

$$\text{averagePowerHalfOrbitSunSeeker} = \frac{\text{energyHalfOrbitSunSeeker}}{\text{halfOrbitSeconds}} \quad \text{(* watts *)}$$

Out[29]=



Out[30]= 3208.52

Out[31]= 1.18834